Pensieve header: $\mathcal A$-calculus and the Hodge star. Continues Alpha.nb.

$\mathcal{A}$-Calculus

In[∘]:=
```
WP[Wedge[u___], Wedge[v___]] := Signature[{u, v}] * Wedge @@ Sort[{u, v}];
WP[0, _] = WP[_, 0] = 0;
WP[A_, B_] :=
  Expand[Distribute[A ** B] /. (a_. * u_Wedge) ** (b_. * v_Wedge) :→ a b WP[u, v]];
```

In[∘]:=
```
WP[Wedge[] + Wedge[a] - 2 b ∧ a, Wedge[] - 3 Wedge[b] + 7 c ∧ d]
```

Out[∘]=

Wedge[] + Wedge[a] − 3 Wedge[b] − a ∧ b + 7 c ∧ d + 7 a ∧ c ∧ d + 14 a ∧ b ∧ c ∧ d

In[∘]:=
```
WExp[A_] := Module[{s = Wedge[], t = Wedge[], k = 0},
  While[t =!= 0, s += (t = Expand[WP[t, A] / (++k)])]; s]
```

In[∘]:=
```
WExp[a ∧ b + c ∧ d + e ∧ f]
```

Out[∘]=

Wedge[] + a ∧ b + c ∧ d + e ∧ f + a ∧ b ∧ c ∧ d + a ∧ b ∧ e ∧ f + c ∧ d ∧ e ∧ f + a ∧ b ∧ c ∧ d ∧ e ∧ f

In[∘]:=
```
c_{x_,y_}[w_Wedge] := Module[{i, j},
   {i} = FirstPosition[w, x, {0}]; {j} = FirstPosition[w, y, {0}];
   Piecewise[{
     {w,                                                   (i == 0) ∧ (j == 0)},
     {(-1)^(i+j+If[i>j,0,1]) Delete[w, {{i}, {j}}],       (i > 0) ∧ (j > 0)}
   }]
  ];
c_{x_,y_}[ℰ_] := ℰ /. w_Wedge :→ c_{x,y}[w]
```

In[∘]:=
```
WExp[a ∧ b + 2 c ∧ d]
c_{d,c}@WExp[a ∧ b + 2 c ∧ d]
```

Out[∘]=

Wedge[] + a ∧ b + 2 c ∧ d + 2 a ∧ b ∧ c ∧ d

Out[∘]=

− Wedge[] − a ∧ b



$X_{ijkl}[S, T]$ $\bar{X}_{ijkl}[S, T]$ $P_{ij}[T]$

In[∘]:=
```
𝒜[X_{i_,j_,k_,l_}[S_, T_]] := 𝒜[{l, i}, {j, k}, ⟨|ξ_i → S, x_j → T, x_k → S, ξ_l → T|⟩,
   Expand[T^(-1/2) WExp[Expand[{ξ_l, ξ_i}.(1 1-T; 0 T).{x_j, x_k}] /. ξ_a_ x_b_ :→ ξ_a ∧ x_b]]];
𝒜[X_{i_,j_,k_,l_}] := 𝒜[X_{i,j,k,l}[τ_i, τ_l]];
```

In[∘]:=
```
𝒜[X̄_{i_,j_,k_,l_}[S_, T_]] := 𝒜[{i, j}, {k, l}, ⟨|ξ_i → S, ξ_j → T, x_k → S, x_l → T|⟩,
   Expand[T^(1/2) WExp[Expand[{ξ_i, ξ_j}.(T^-1 0; 1-T^-1 1).{x_k, x_l}] /. ξ_a_ x_b_ :→ ξ_a ∧ x_b]]];
𝒜[X̄_{i_,j_,k_,l_}] := 𝒜[X̄_{i,j,k,l}[τ_i, τ_j]];
```

In[∘]:=
```
𝒜[P_{i_,j_}[T_]] := 𝒜[{i}, {j}, ⟨|ξ_i → T, x_j → T|⟩, WExp[ξ_i ∧ x_j]];
𝒜[P_{i_,j_}] := 𝒜[P_{i,j}[τ_i]]
```

In[∘]:= 𝒜[X_{1,2,3,4}[u, v]]

Out[∘]=

$$\mathcal{A}\Big[\{4,1\},\{2,3\},\langle|\xi_1\to u, x_2\to v, x_3\to u, \xi_4\to v|\rangle,\ \frac{\mathrm{Wedge}[\,]}{\sqrt{v}}-\frac{x_2\wedge\xi_4}{\sqrt{v}}-\sqrt{v}\,x_3\wedge\xi_1-\frac{x_3\wedge\xi_4}{\sqrt{v}}+\sqrt{v}\,x_3\wedge\xi_4+\sqrt{v}\,x_2\wedge x_3\wedge\xi_1\wedge\xi_4\Big]$$

Linearity:

In[∘]:=
```
𝒜 /: α_ × 𝒜[is_, os_, cs_, w_] := 𝒜[is, os, cs, Expand[α w]]
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] + 𝒜[is2_, os2_, cs2_, w2_] /;
  (Sort@is1 == Sort@is2) ∧ (Sort@os1 == Sort@os2) ∧
   (Sort@Normal@cs1 == Sort@Normal@cs2) := 𝒜[is1, os1, cs1, w1 + w2]
```

Deciding if two $\calA$'s are equal:

In[∘]:=
```
𝒜 /: 𝒜[is1_, os1_, _, w1_] ≡ 𝒜[is2_, os2_, _, w2_] :=
 TrueQ[(Sort@is1 === Sort@is2) ∧ (Sort@os1 === Sort@os2) ∧ PowerExpand[w1 == w2]]
```

Disjoint unions:

In[∘]:=
```
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] 𝒜[is2_, os2_, cs2_, w2_] :=
 𝒜[is1 ⋃ is2, os1 ⋃ os2, Join[cs1, cs2], WP[w1, w2]]
```

In[∘]:=
```
Short[𝒜[X2,4,3,1[S, T]] 𝒜[X̄3,4,6,5], 5]
```

Out[∘]//Short=
```
𝒜[{1, 2, 3, 4}, {3, 4, 5, 6}, <|ξ2 → S, x4 → T, x3 → S, ξ1 → T, ξ3 → τ3, ξ4 → τ4, x6 → τ3, x5 → τ4|>,
 √τ4 Wedge[]/√T - √τ4 x3∧ξ1/√T + √T √τ4 x3∧ξ1 - √T √τ4 x3∧ξ2 - √τ4 x4∧ξ1/√T -
 √τ4 x5∧ξ4/√T - x6∧ξ3/(√T √τ4) + x6∧ξ4/(√T √τ4) - √τ4 x6∧ξ4/√T + √T √τ4 x3∧x4∧ξ1∧ξ2 -
 √τ4 x3∧x5∧ξ1∧ξ4/√T + <<21>> + x4∧x6∧ξ1∧ξ4/(√T √τ4) - √τ4 x4∧x6∧ξ1∧ξ4/√T + x5∧x6∧ξ3∧ξ4/(√T √τ4) -
 √T √τ4 x3∧x4∧x5∧ξ1∧ξ2∧ξ4 - √T x3∧x4∧x6∧ξ1∧ξ2∧ξ3/√τ4 + √T x3∧x4∧x6∧ξ1∧ξ2∧ξ4/√τ4 -
 √T √τ4 x3∧x4∧x6∧ξ1∧ξ2∧ξ4 - x3∧x5∧x6∧ξ1∧ξ3∧ξ4/(√T √τ4) + √T x3∧x5∧x6∧ξ1∧ξ3∧ξ4/√τ4 -
 √T x3∧x5∧x6∧ξ2∧ξ3∧ξ4/√τ4 - x4∧x5∧x6∧ξ1∧ξ3∧ξ4/(√T √τ4) + √T x3∧x4∧x5∧x6∧ξ1∧ξ2∧ξ3∧ξ4/√τ4]
```

In[∘]:=
```
c_{h_,t_}@𝒜[is_, os_, cs_, w_] := 𝒜[
    DeleteCases[is, t], DeleteCases[os, h], KeyDrop[cs, {x_h, ξ_t}], c_{x_h,ξ_t}[w]
  ] /. If[MatchQ[cs[ξ_t], τ_], cs[ξ_t] → cs[x_h], cs[x_h] → cs[ξ_t]];
```

In[∘]:=
```
c4,4[𝒜[X2,4,3,1[S, T]] 𝒜[X̄3,4,6,5]]
```

Out[∘]=
```
𝒜[{1, 2, 3}, {3, 5, 6}, <|ξ2 → S, x3 → S, ξ1 → T, ξ3 → τ3, x6 → τ3, x5 → T|>,
 Wedge[] - x3∧ξ1 + T x3∧ξ1 - T x3∧ξ2 - x5∧ξ1 - x6∧ξ1 + x6∧ξ1/T - x6∧ξ3/T +
 T x3∧x5∧ξ1∧ξ2 - x3∧x6∧ξ1∧ξ2 + T x3∧x6∧ξ1∧ξ2 + x3∧x6∧ξ1∧ξ3 -
 x3∧x6∧ξ1∧ξ3/T - x3∧x6∧ξ2∧ξ3 - x5∧x6∧ξ1∧ξ3/T - x3∧x5∧x6∧ξ1∧ξ2∧ξ3]
```

In[∘]:=
```
c@𝒜[is_, os_, cs_, w_] := Fold[c_{#2,#2}[#1] &, 𝒜[is, os, cs, w], is ⋂ os]
𝒜[{A_𝒜}] := c[A];
𝒜[{A1_𝒜, As__𝒜}] := Module[{A2},
  A2 = First@MaximalBy[{As}, Length[A1〚1〛 ⋂ #〚2〛] + Length[A1〚2〛 ⋂ #〚1〛] &];
  𝒜[Join[{c[A1 A2]}, DeleteCases[{As}, A2]]]]
𝒜[𝒪s_List] := 𝒜[𝒜 /@ 𝒪s]
```

In[∘]:=
```
c[𝒜[X2,4,3,1[S, T]] 𝒜[X̄3,4,6,5]]
```

Out[∘]=
```
𝒜[{1, 2}, {5, 6}, <|ξ2 → S, ξ1 → T, x6 → S, x5 → T|>, Wedge[] - x5∧ξ1 - x6∧ξ2 - x5∧x6∧ξ1∧ξ2]
```

In[∘]:= $\mathcal{A}$@$\{\mathcal{A}[X_{2,4,3,1}[S, T]], \mathcal{A}[\overline{X}_{3,4,6,5}]\}$

Out[∘]=

$\mathcal{A}[\{1, 2\}, \{5, 6\}, \langle| \xi_2 \to S, \xi_1 \to T, x_6 \to S, x_5 \to T |\rangle, \text{Wedge}[\,] - x_5 \wedge \xi_1 - x_6 \wedge \xi_2 - x_5 \wedge x_6 \wedge \xi_1 \wedge \xi_2]$

Skein Relations



In[∘]:= $\mathcal{A}$@$\{\overline{X}_{4,1,6,3}[v, u], \overline{X}_{3,2,5,4}\}$

Out[∘]=

$$\mathcal{A}\Big[\{1, 2\}, \{5, 6\}, \langle| \xi_2 \to v, x_5 \to u, \xi_1 \to u, x_6 \to v |\rangle,$$
$$\sqrt{u}\,\sqrt{v}\,\text{Wedge}[\,] - \frac{\sqrt{u}\, x_5 \wedge \xi_1}{\sqrt{v}} + \frac{\sqrt{u}\, x_5 \wedge \xi_2}{\sqrt{v}} - \sqrt{u}\,\sqrt{v}\, x_5 \wedge \xi_2 + \frac{\sqrt{v}\, x_6 \wedge \xi_1}{\sqrt{u}} - \sqrt{u}\,\sqrt{v}\, x_6 \wedge \xi_1 -$$
$$\frac{\sqrt{v}\, x_6 \wedge \xi_2}{\sqrt{u}} - \frac{\sqrt{u}\, x_5 \wedge x_6 \wedge \xi_1 \wedge \xi_2}{\sqrt{v}} - \frac{\sqrt{v}\, x_5 \wedge x_6 \wedge \xi_1 \wedge \xi_2}{\sqrt{u}} + \sqrt{u}\,\sqrt{v}\, x_5 \wedge x_6 \wedge \xi_1 \wedge \xi_2\Big]$$

Reidemeister 2



In[∘]:= $\{\mathcal{A}$@$\{X_{2,4,3,1}[S, T], \overline{X}_{3,4,6,5}\} \equiv \mathcal{A}$@$\{P_{1,5}[T], P_{2,6}[S]\},$
$\mathcal{A}$@$\{\overline{X}_{3,1,2,4}[S, T], X_{6,5,3,4}\} \equiv \mathcal{A}$@$\{P_{1,5}[T], P_{6,2}[S]\}\}$

Out[∘]=

{True, True}

Reidemeister 3



In[∘]:= $\mathcal{A}$@{$X_{2,5,4,1}[T_2, T_1]$, $X_{3,7,6,5}[T_3, T_1]$, $X_{6,9,8,4}$} ≡ $\mathcal{A}$@{$X_{3,5,4,2}[T_3, T_2]$, $X_{4,6,8,1}[T_3, T_1]$, $X_{5,7,9,6}$}

Out[∘]=

True

Reidemeister 1



In[∘]:= {$\mathcal{A}$@{$X_{3,3,2,1}$} ≡ $\tau_1^{-1/2}$ $\mathcal{A}$@{$P_{1,2}$}, $\mathcal{A}$@{$X_{1,2,3,3}$} ≡ $\tau_1^{1/2}$ $\mathcal{A}$@{$P_{1,2}$},
$\mathcal{A}$@{$\overline{X}_{1,3,3,2}$} ≡ $\tau_1^{-1/2}$ $\mathcal{A}$@{$P_{1,2}$}, $\mathcal{A}$@{$\overline{X}_{3,1,2,3}$} ≡ $\tau_1^{1/2}$ $\mathcal{A}$@{$P_{1,2}$}}

Out[∘]=

{True, True, True, True}

The Relation with the Multivariable Alexander Polynomial



In[∘]:= MVA = $u^{-1/2} v^{-1/2} w^{-1/2}$ (u − 1) (v − 1) (w − 1);

In[∘]:= A = {$\overline{X}_{1,12,2,13}[u, v]$, $\overline{X}_{13,2,6,3}$, $X_{8,4,9,3}$, $X_{4,10,5,9}$, $X_{6,17,7,16}[v, w]$,
$X_{15,8,16,7}$, $\overline{X}_{14,10,15,11}$, $\overline{X}_{11,17,12,14}$} // $\mathcal{A}$ // Last // Factor

Out[∘]=

$$\frac{(-1+u)^2\,(-1+v)\,(-1+w)\,(\text{Wedge}[\,] - x_5 \wedge \xi_1)}{u\,v}$$

In[∘]:= A == $u^{-1/2}$ (u − 1) u^0 $v^{-1/2}$ $w^{1/2}$ MVA (Wedge[] − $x_5 \wedge \xi_1$)

Out[∘]=

True

Overcrossings Commute but Undercrossings don’t



In[∘]:= $\{\mathcal{A}@\{X_{2,7,5,1}, X_{3,4,6,7}\} \equiv \mathcal{A}@\{X_{3,7,6,1}, X_{2,4,5,7}\},\ \mathcal{A}@\{\overline{X}_{1,2,7,5}, \overline{X}_{7,3,4,6}\} \equiv \mathcal{A}@\{\overline{X}_{1,3,7,6}, \overline{X}_{7,2,4,5}\}\}$

Out[∘]=

{True, False}

The Conway Relation



In[∘]:= $\mathcal{A}@\{X_{2,3,4,1}[T, T]\} - \mathcal{A}@\{\overline{X}_{1,2,3,4}[T, T]\} \equiv (T^{-1/2} - T^{1/2})\ \mathcal{A}@\{P_{1,4}[T], P_{2,3}[T]\}$

Out[∘]=

True

Conway's Second Set of Identities



In[∘]:= $\{\mathcal{A}@\{X_{2,4,3,1}[v, u], X_{4,6,5,3}\} + \mathcal{A}@\{\overline{X}_{1,2,4,3}[u, v], \overline{X}_{3,4,6,5}\} \equiv$
$(u^{1/2}\, v^{1/2} + u^{-1/2}\, v^{-1/2})\ \mathcal{A}@\{P_{1,5}[u], P_{2,6}[v]\},$
$\mathcal{A}@\{\overline{X}_{4,1,6,3}[v, u], \overline{X}_{3,2,5,4}\} + \mathcal{A}@\{X_{1,6,3,4}[u, v], X_{2,5,4,3}\} \equiv$
$(u^{1/2}\, v^{-1/2} + u^{-1/2}\, v^{1/2})\ \mathcal{A}@\{P_{1,5}[u], P_{2,6}[v]\}\}$

Out[∘]=

{True, True}



In[∘]:= $\{\mathcal{A}@\{X_{2,3,4,1}\} + \mathcal{A}@\{\overline{X}_{2,1,4,3}\} \equiv (\tau_1^{1/2} + \tau_1^{-1/2})\ \mathcal{A}@\{P_{1,3}, P_{2,4}\},$
$\mathcal{A}@\{\overline{X}_{1,2,3,4}\} + \mathcal{A}@\{X_{1,4,3,2}\} \equiv (\tau_2^{1/2} + \tau_2^{-1/2})\ \mathcal{A}@\{P_{1,3}, P_{2,4}\}\}$

Out[∘]=

{True, True}

Conway's Third Identity and Virtual Version (Archibald)



In[∘]:= $\mathcal{A}$@$\{X_{6,4,9,1}, \overline{X}_{4,5,7,8}, \overline{X}_{2,3,5,6}\}$ + $\mathcal{A}$@$\{X_{2,4,5,1}, \overline{X}_{4,3,7,6}, X_{6,8,9,5}\}$ ≡ $\mathcal{A}$@$\{\overline{X}_{1,6,4,9}, X_{5,7,8,4}, X_{3,5,6,2}\}$ + $\mathcal{A}$@$\{\overline{X}_{1,2,4,5}, X_{3,7,6,4}, \overline{X}_{5,6,8,9}\}$

Out[∘]=

True



In[∘]:= $\mathcal{A}$@$\{X_{3,7,6,1}, \overline{X}_{7,2,4,5}\}$ + $\mathcal{A}$@$\{X_{2,4,7,1}, X_{3,5,6,7}\}$ ≡ $\mathcal{A}$@$\{X_{3,7,6,2}, X_{7,4,5,1}\}$ + $\mathcal{A}$@$\{\overline{X}_{1,2,7,5}, X_{3,4,6,7}\}$

Out[∘]=

True

Jun Murakami’s Fifth Axiom and Virtual Versions (Archibald)



In[∘]:= $\mathcal{A}$@$\{X_{1,4,2,5}[T, S], X_{4,3,5,2}\}$ ≡ $\frac{\sqrt{S}\ (1-T)}{\sqrt{T}}$ $\mathcal{A}$@$\{P_{1,3}[T]\}$

Out[∘]=

True



In[∘]:= $\mathcal{A}$@$\{X_{3,2,3,1}[S, T]\}$ ≡ $\left(T^{-1/2} - T^{1/2}\right)$ $\mathcal{A}$@$\{P_{1,2}[T]\}$

Out[∘]=

True

In[∘]:= $\mathcal{A}$@{$X_{1,3,2,3}$}

Out[∘]=

$\mathcal{A}[\{1\},\{2\},\langle|\xi_1\to\tau_1,\ x_2\to\tau_1|\rangle,0]$

Jun Murakami's Third Axiom



In[∘]:=

$\mathcal{A}_{2112}=\mathcal{A}@\{X_{3,8,7,2},X_{7,10,9,1},X_{10,11,4,9},X_{8,6,5,11}\};$

$\mathcal{A}_{1221}=\mathcal{A}@\{X_{2,8,7,1},X_{3,10,9,8},X_{10,6,11,9},X_{11,5,4,7}\};$

$\mathcal{A}_{2211}=\mathcal{A}@\{X_{3,8,7,2},X_{8,6,9,7},X_{9,11,10,1},X_{11,5,4,10}\};$

$\mathcal{A}_{1122}=\mathcal{A}@\{X_{2,8,7,1},X_{8,9,4,7},X_{3,11,10,9},X_{11,6,5,10}\};$

$\mathcal{A}_{11}=\mathcal{A}@\{X_{2,8,7,1},X_{8,5,4,7},P_{3,6}\};\quad \mathcal{A}_{22}=\mathcal{A}@\{X_{3,8,7,2},X_{8,6,5,7},P_{1,4}\};$

$\mathcal{A}_{\phi}=\mathcal{A}@\{P_{1,4},P_{2,5},P_{3,6}\};$

$g_+[z\_]:=z^{1/2}+z^{-1/2};\quad g_-[z\_]:=z^{1/2}-z^{-1/2};$

$g_+[\tau_1]\,g_-[\tau_2]\,\mathcal{A}_{2112}-g_-[\tau_2]\,g_+[\tau_3]\,\mathcal{A}_{1221}-g_-[\tau_3/\tau_1]\,(\mathcal{A}_{2211}+\mathcal{A}_{1122})+$
$\quad g_-[\tau_2\,\tau_3/\tau_1]\,g_+[\tau_3]\,\mathcal{A}_{11}-g_+[\tau_1]\,g_-[\tau_1\,\tau_2/\tau_3]\,\mathcal{A}_{22}\equiv g_-\left[\tau_3^2/\tau_1^2\right]\mathcal{A}_{\phi}$

Out[∘]=

True

The Naik-Stanford Double Delta Move and Virtual Versions (Archibald)



In[∘]:=

$$\text{Timing}\Big[\mathcal{A}@\{X_{6,10,28,24}[w,v], \overline{X}_{28,3,29,19}[w,v], X_{26,20,27,19}[w,v], \overline{X}_{27,23,11,24}[w,v], X_{1,12,13,30}[u,w], \overline{X}_{13,5,14,25}[u,w], X_{17,26,18,25}[u,w], \overline{X}_{18,29,8,30}[u,w], X_{4,7,22,15}[v,u], \overline{X}_{22,2,23,16}[v,u], X_{20,17,21,16}[v,u], \overline{X}_{21,14,9,15}[v,u]\} \equiv \mathcal{A}@\{X_{5,9,25,21}[w,v], \overline{X}_{25,4,26,22}[w,v], X_{29,23,30,22}[w,v], \overline{X}_{30,20,12,21}[w,v], X_{2,11,16,27}[u,w], \overline{X}_{16,6,17,28}[u,w], X_{14,29,15,28}[u,w], \overline{X}_{15,26,7,27}[u,w], X_{3,8,19,18}[v,u], \overline{X}_{19,1,20,13}[v,u], X_{23,14,24,13}[v,u], \overline{X}_{24,17,10,18}[v,u]\}\Big]$$

Out[∘]=

{198.719, True}



In[∘]:=

$$\mathcal{A}@\{X_{1,8,11,3}[u,v], \overline{X}_{11,2,12,7}[u,v], X_{12,10,13,4}[u,w], \overline{X}_{13,5,6,9}[u,w]\} \equiv \mathcal{A}@\{X_{1,10,11,4}[u,w], \overline{X}_{11,5,12,9}[u,w], X_{12,8,13,3}[u,v], \overline{X}_{13,2,6,7}[u,v]\}$$

Out[∘]=

True



In[∘]:=

$$\mathcal{A}@\{\overline{X}_{20,1,10,13}[v,u], X_{3,14,19,13}[v,u], X_{14,11,15,21}[u,w], \overline{X}_{15,6,7,22}[u,w], X_{2,12,16,22}[u,w], \overline{X}_{16,5,17,21}[u,w], \overline{X}_{19,17,9,18}[v,u], X_{4,8,20,18}[v,u]\} \equiv \mathcal{A}@\{X_{1,11,13,21}[u,w], \overline{X}_{13,6,14,22}[u,w], \overline{X}_{20,14,10,15}[v,u], X_{3,7,19,15}[v,u], \overline{X}_{19,2,9,16}[v,u], X_{4,17,20,16}[v,u], X_{17,12,18,22}[u,w], \overline{X}_{18,5,8,21}[u,w]\}$$

Out[∘]=

True

Γ ↔ 𝒜 Conversions

pdf

In[∘]:=
```
Γ@𝒜[is_, os_, cs_, w_] := Module[{i, j, ω = Coefficient[w, Wedge[]]},
   Γ[is, os, cs, ω, Sum[Cancel[-Coefficient[w, x_j ∧ ξ_i] ξ_i x_j / ω], {i, is}, {j, os}]]
  ];
𝒜@Γ[is_, os_, cs_, ω_, λ_] := 𝒜[is, os, cs, Expand[ω WExp[Expand[λ] /. ξ_a_ x_b_ :> ξ_a ∧ x_b]]];
```

tex

The conversions are inverses of each other:

pdf

In[∘]:=
```
γ = Γ[{1, 2, 3}, {1, 2, 3}, {x_1 → τ_1, x_2 → τ_2, x_3 → τ_3, ξ_1 → τ_1, ξ_2 → τ_2, ξ_3 → τ_3}, ω,
   a_11 x_1 ξ_1 + a_12 x_2 ξ_1 + a_13 x_3 ξ_1 + a_21 x_1 ξ_2 + a_22 x_2 ξ_2 + a_23 x_3 ξ_2 + a_31 x_1 ξ_3 + a_32 x_2 ξ_3 + a_33 x_3 ξ_3];
Γ@𝒜@γ == γ
```

Out[∘]=
pdf

True

The Slant Product

In[∘]:=
```
Unprotect[Conjugate];
(T^p_.)* := T^-p;
Protect[Conjugate];
```

In[∘]:=
```
WP[Wedge[u___], Wedge[v___]] := Signature[{u, v}] * Wedge @@ Sort[{u, v}];
WP[0, _] = WP[_, 0] = 0;
WP[A_, B_] :=
  Expand[Distribute[A ** B] /. (a_. * u_Wedge) ** (b_. * v_Wedge) :> a b WP[u, v]];
```

In[∘]:=
```
ep[a_, b_] := If[a === b, 1, 0];
```

In[∘]:=
```
Slant_p_[u_Wedge, v_Wedge] := Module[{n = Length@u, k = Length@v},
   If[n < k, 0,
    Sum[
     Signature[σ] * Times @@ MapThread[p, {List @@ u[[σ]], List @@ v}] * Delete[u, List /@ σ],
     {σ, Permutations[Range@n, {k}]}]
   ]
  ];
Slant_[0, _] = Slant_[_, 0] = 0;
Slant_p_[A_, B_] :=
  Expand[Distribute[A ** B] /. (a_. * u_Wedge) ** (b_. * v_Wedge) :> a b* Slant_p[u, v]];
```

In[∘]:=
```
Slant_ep[Wedge[1, 2, 3, 4], Wedge[4, 2] + 7 Wedge[4, 1, 2] + T Wedge[2]]
```

Out[∘]=

$$7\,\text{Wedge}[3] - 1\wedge 3 + \frac{1\wedge 3\wedge 4}{T}$$

The Pairings

In[∘]:=
```
p_ς_[c_, d_] := Expand@Cancel[T^(1/2) (T - 1)^-1 Module[{e, f},
    Expand[c (d /. {T → T*, ξ_i_ ↦ ξ_i*, x_i_ ↦ x_i*})] /.
     {t_i_* t_i_ → 0, x_i_* x_i_ → T - T*, (f : ξ | x)_j_* (e : ξ | x)_i_ ↦
       If[Position[ς, e_i][[1, 1]] < Position[ς, f_j][[1, 1]], T - 1, 1 - T*]}
   ]];
```



In[∘]:=
```
𝒜[X_1,2,3,4[T, T]]
```

Out[∘]=

$$\mathcal{A}\Big[\{4, 1\}, \{2, 3\}, \langle| \xi_1 \to T, x_2 \to T, x_3 \to T, \xi_4 \to T |\rangle, \frac{\text{Wedge}[\,]}{\sqrt{T}} - \frac{x_2\wedge\xi_4}{\sqrt{T}} - \sqrt{T}\, x_3\wedge\xi_1 - \frac{x_3\wedge\xi_4}{\sqrt{T}} + \sqrt{T}\, x_3\wedge\xi_4 + \sqrt{T}\, x_2\wedge x_3\wedge\xi_1\wedge\xi_4\Big]$$

In[∘]:=
```
ς0 = {ξ_1, x_2, x_3, ξ_4};
Slant_p_ς0[Wedge @@ ς0, Last@𝒜[X_1,2,3,4[T, T]]]
```

Out[∘]=

$$\frac{\text{Wedge}[\,]}{T^{5/2}} - \frac{\text{Wedge}[\,]}{T^{3/2}} + \frac{x_2\wedge x_3}{T^{3/2}} - \frac{x_2\wedge x_3}{\sqrt{T}} + \frac{x_2\wedge\xi_4}{T^{3/2}} - \frac{x_2\wedge\xi_4}{\sqrt{T}} +$$
$$2\sqrt{T}\, x_2\wedge\xi_4 - T^{3/2}\, x_2\wedge\xi_4 + \sqrt{T}\, x_3\wedge\xi_4 + \frac{\xi_1\wedge x_2}{T^{3/2}} - \frac{3\,\xi_1\wedge x_2}{\sqrt{T}} + \sqrt{T}\,\xi_1\wedge x_2 + \frac{\xi_1\wedge x_3}{T^{3/2}} -$$
$$\frac{2\,\xi_1\wedge x_3}{\sqrt{T}} + \frac{\xi_1\wedge\xi_4}{T^{3/2}} - \frac{\xi_1\wedge\xi_4}{\sqrt{T}} + \sqrt{T}\,\xi_1\wedge\xi_4 - T^{3/2}\,\xi_1\wedge\xi_4 + \sqrt{T}\,\xi_1\wedge x_2\wedge x_3\wedge\xi_4$$

In[ ]:=
$$(T^{-5/2}-T^{-3/2})$$
$$\text{WExp@Expand}\Big[(T^{-5/2}-T^{-3/2})^{-1}\Big(\frac{x_2\wedge x_3}{T^{3/2}}-\frac{x_2\wedge x_3}{\sqrt{T}}+\frac{x_2\wedge \xi_4}{T^{3/2}}-\frac{x_2\wedge \xi_4}{\sqrt{T}}+2\sqrt{T}\,x_2\wedge\xi_4-T^{3/2}x_2\wedge\xi_4+\sqrt{T}\,x_3\wedge\xi_4+\frac{\xi_1\wedge x_2}{T^{3/2}}-\frac{3\,\xi_1\wedge x_2}{\sqrt{T}}+\sqrt{T}\,\xi_1\wedge x_2+\frac{\xi_1\wedge x_3}{T^{3/2}}-\frac{2\,\xi_1\wedge x_3}{\sqrt{T}}+\frac{\xi_1\wedge \xi_4}{T^{3/2}}-\frac{\xi_1\wedge \xi_4}{\sqrt{T}}+\sqrt{T}\,\xi_1\wedge\xi_4-T^{3/2}\xi_1\wedge\xi_4\Big)\Big] \text{ // Simplify // Expand}$$

Out[ ]=
$$\frac{\text{Wedge}[\,]}{T^{5/2}}-\frac{\text{Wedge}[\,]}{T^{3/2}}+\frac{x_2\wedge x_3}{T^{3/2}}-\frac{x_2\wedge x_3}{\sqrt{T}}-\frac{x_2\wedge \xi_1}{T^{3/2}}+\frac{3\,x_2\wedge \xi_1}{\sqrt{T}}-\sqrt{T}\,x_2\wedge\xi_1+\frac{x_2\wedge \xi_4}{T^{3/2}}-\frac{x_2\wedge \xi_4}{\sqrt{T}}+2\sqrt{T}\,x_2\wedge\xi_4-T^{3/2}x_2\wedge\xi_4-\frac{x_3\wedge \xi_1}{T^{3/2}}+\frac{2\,x_3\wedge \xi_1}{\sqrt{T}}+\sqrt{T}\,x_3\wedge\xi_4+\frac{\xi_1\wedge \xi_4}{T^{3/2}}-\frac{\xi_1\wedge \xi_4}{\sqrt{T}}+\sqrt{T}\,\xi_1\wedge\xi_4-T^{3/2}\xi_1\wedge\xi_4+\sqrt{T}\,x_2\wedge x_3\wedge\xi_1\wedge\xi_4$$